

Chapter 6

Integrals

In the first section, we use Maple's `leftbox` and `rightbox` commands to visualize the process of approximating the area under a curve. In the second section, we compute some Riemann sums with the help of Maple's `Sum` command. The integration command `Int` is also introduced. The next three sections demonstrate how Maple can be used to perform integration by substitution, integration by parts, and integration using a partial fraction decomposition. The final section discusses the midpoint rule, the trapezoidal rule and Simpson's rule for approximating definite integrals.

Since Maple has a built-in integration command, it is reasonable to wonder why we need to discuss techniques of integration. It turns out that Maple cannot integrate everything, (Try integrating $\int x \ln(x + \sqrt{1 + x^2}) dx$.) but with a little help from us, the number of integrands that Maple is able to integrate in closed form increases. (By the end of the chapter you will compute $\int x \ln(x + \sqrt{1 + x^2}) dx$ in Exercise 14.)

6.1 Visualizing Riemann Sums

Let's compute the area of the region under the graph of the function $f(x) = x^2$ over the interval $1 \leq x \leq 3$. We enter this as the Maple function:


```
> f:=x->x^2:
```

First, divide this interval into n subintervals of equal length, $\Delta x = \frac{3-1}{n} = \frac{2}{n}$. Here, n is typically a large number such as 10 or 100 (later, n will approach ∞). The area under the graph of $f(x) = x^2$ is approximated by the sum of the areas of n rectangles where the base of a typical rectangle is one of the n subintervals and the height is the value of the function $f(x)$ at the left or right endpoint of the subinterval. To get a picture with $n = 10$ rectangles, load the `student` and `plots` packages and enter the `leftbox` and `rightbox` commands as follows:

```

> with(student): with(plots):
> Lplot:=leftbox(f(x), x=1..3, 10, xtickmarks=3,
> title="Left Riemann Sum"):
> Rplot:=rightbox(f(x), x=1..3, 10, xtickmarks=3,
> title="Right Riemann Sum"):
> display(array([Lplot,Rplot]));

```



The `leftbox(f(x),x=1..3,10)` command draws the graph of $f(x)$ over the interval $1 \leq x \leq 3$, together with 10 rectangles; where the height of each rectangle is the value of the function at the left endpoint of the base of the rectangle. Similarly, the `rightbox` command takes the height of each rectangle as the value of the function at the right endpoint of the base. (The `display` command with an `array` argument shows the two plots next to each other instead of superimposing them.) The sum of the areas of these rectangles is called a left (right) Riemann sum approximation to the area under the graph of f .

Since the graph of f is increasing over the interval $1 \leq x \leq 3$, the left Riemann sum is less than the area under the graph of f (see the picture above) and is called a lower bound; while the right Riemann sum is more than the area under the graph and is called an upper bound.


The value $n = 10$ can be changed to any positive integer. Try the `leftbox` and `rightbox` commands with $n = 20, 50$ and 100 . Note that, as n gets larger, the total area of the rectangles more closely approximates the area under the graph of f with the left Riemann sum as a lower estimate and the right Riemann sum as an upper estimate.

You can see this better with an animation. The `ApproximateInt` command in the `Student[Calculus1]` package allows you to animate Riemann sums. For example, to see a left Riemann sum with 4, 8, 16, 32 and 64 intervals execute

```

> with(Student[Calculus1]):
> ApproximateInt(x^2, 1..3, output=animation, iterations=5,
> method=left, partition=4, subpartition=all, refinement=halve);

```




Try this yourself, since you cannot see an animation in a book. To animate it, click in the plot and click on the PLAY button which is the right pointing triangle (\triangleright) on the plot toolbar. You can also access this animation as a Maplet accessible in the Standard interface from the menu TOOLS > TUTORS > CALCULUS - SINGLE VARIABLE > RIEMANN SUMS.

This same rectangle construction can be applied to any interval $a \leq x \leq b$. Try different values of a and b (and n). Note that, for negative values of a and b , f is a decreasing function on $a \leq x \leq b$. Therefore the left Riemann sums are upper estimates and the right Riemann sums are lower estimates.

6.2 The Computation of Integrals

We continue with the preceding example of the integral of $f(x) = x^2$ for $a \leq x \leq b$. and we look at a right Riemann sum.



As in the previous section, the interval $a \leq x \leq b$ is divided into n subintervals, each of length

$$\Delta x = \frac{b - a}{n}$$

Let x_i denote the right endpoint of the i^{th} subinterval, where i is a counting index that runs from 1 to n . The area of the i^{th} box is the product of the

interval length Δx and the height of the box, which is $f(x_i)$. So the sum of the areas of the n rectangles is

$$\sum_{i=1}^n f(x_i) \Delta x$$

Before using Maple to compute this sum, a formula for x_i must be derived. Note the following pattern: $x_1 = a + \Delta x$, $x_2 = a + 2\Delta x$, $x_3 = a + 3\Delta x$ and so forth. So in general

$$x_i = a + i\Delta x$$

Now these formulas can be entered into Maple. First, enter f as a function.

```
> f:=x->x^2;
```

$$f := x \rightarrow x^2$$

Then enter Δx (labeled Dx in the Maple session) and x_i :

```
> Dx:=(b-a)/n; x[i]:=a+i*Dx;
```

$$Dx := \frac{b-a}{n}$$

$$x_i := a + \frac{i(b-a)}{n}$$

The above sum is entered with Maple's `Sum` command.

```
> riemsum:=Sum(f(x[i])*Dx,i=1..n);
```

$$riemsum := \sum_{i=1}^n \frac{(a + \frac{i(b-a)}{n})^2 (b-a)}{n}$$

This sum can also be displayed without computing Δx and x_i by using the `rightsum` command from the student package.

```
> with(student):
```

```
> rightsum(f(x),x=a..b,n);
```

$$\frac{(b-a) \left(\sum_{i=1}^n (a + \frac{i(b-a)}{n})^2 \right)}{n}$$

The `Sum` command with an upper case `S` and the `rightsum` command display the sum without calculating its value, as illustrated above. This allows the sum to be checked for typing errors. The `value(%)` command can be added on the same line to evaluate this sum. Since the result is long, we suppress the output of `value(%)`. However it is significantly simplified if we `expand` the result:

```
> riemsum:=Sum(f(x[i])*Dx,i=1..n); value(%):
```

$$riemsum := \sum_{i=1}^n \frac{(a + \frac{i(b-a)}{n})^2 (b-a)}{n}$$

```
> riemsum:=expand(%);
```

$$\text{riemsum} := -\frac{ab^2}{2n} - \frac{a^3}{3} - \frac{a^2b}{2n} + \frac{a^2b}{2n^2} - \frac{ab^2}{2n^2} + \frac{b^3}{3} + \frac{a^3}{2n} - \frac{a^3}{6n^2} + \frac{b^3}{2n} + \frac{b^3}{6n^2}$$

This output represents the sum of the areas of the n rectangles that approximates the area under the graph of $f(x) = x^2$ for $a \leq x \leq b$. To get the precise area under the graph, take the limit of this expression as n approaches infinity.

```
> Limit(riemsum,n=infinity); area:=value(%);
```

$$\lim_{n \rightarrow \infty} -\frac{ab^2}{2n} - \frac{a^3}{3} - \frac{a^2b}{2n} + \frac{a^2b}{2n^2} - \frac{ab^2}{2n^2} + \frac{b^3}{3} + \frac{a^3}{2n} - \frac{a^3}{6n^2} + \frac{b^3}{2n} + \frac{b^3}{6n^2}$$

$$\text{area} := -\frac{a^3}{3} + \frac{b^3}{3}$$

This limit of the sum of the areas of rectangles (as the number of rectangles tends to infinity) is the definition of the integral of f from $x = a$ to $x = b$, and is denoted by the symbol

$$\int_a^b f(x) dx$$

For a nonnegative function, such as $f(x) = x^2$, the integral measures the area under the curve. However, the definition of the integral holds even for functions which are not nonnegative.

Maple has a command that integrates expressions directly. (Since \mathbf{f} has been entered as a function, we must evaluate it to get the expression $\mathbf{f}(x)$.) To integrate $\mathbf{f}(x)$, you enter

```
> Int(f(x),x=a..b); area:=value(%);
```

$$\int_a^b x^2 dx$$

$$\text{area} := -\frac{a^3}{3} + \frac{b^3}{3}$$

Like the `Sum` and `Diff` commands, the `Int` command with an uppercase `I` displays the integral without evaluating it, so that it can be checked for typing errors. It is evaluated by adding the `value(%);` command.

Notice that the resulting area is the same as

$$F(b) - F(a) \quad \text{where} \quad F(x) = \frac{x^3}{3}.$$

Further, note that $F(x) = \frac{x^3}{3}$ is an antiderivative of $f(x) = x^2$. This is an illustration of the *Fundamental Theorem of Calculus*, which states that the definite integral $\int_a^b f(x) dx$ (defined as the limit of the sum of areas of rectangles) is the same as $F(b) - F(a)$, where F is an antiderivative of f .

As a further illustration of the Fundamental Theorem, try repeating the above procedure with the functions $f(x) = x^3$ and $f(x) = x^4$. The only Maple command that you must change is the one that involves the definition of the function f . The other statements can be re-executed without change.

The antiderivative of f (or indefinite integral) can also be evaluated.

```
> Int(f(x), x); value(%);
```

$$\int x^2 dx$$

$$\frac{x^3}{3}$$

Note that Maple *does not insert the constant of integration*.

If f is defined as an expression rather than a function (via `f:=x^2`;) the same syntax as above is used except that `f` is typed instead of `f(x)`. That is, `Int(f, x=a..b)`; will display the definite integral $\int_a^b x^2 dx$, while `Int(f, x)`; will display the indefinite integral $\int x^2 dx$ and then `value(%)` will evaluate them.

Maple cannot find an antiderivative for every expression. For example, try integrating $\sqrt{x^5+1}e^x$ with Maple. Some definite integrals have to be approximated either by adding up rectangles or by other more sophisticated techniques discussed later in this chapter. Maple makes it easy to evaluate an approximation to a definite integral by using `evalf`. For example, the integral $\int_1^2 \sqrt{x^5+1}e^x dx$ can be approximated by entering:

```
> Int(sqrt(x^5+1)*exp(x), x=1..2); evalf(%);
```

$$\int_1^2 \sqrt{x^5+1}e^x dx$$

$$16.38809099$$

6.3 Integration by Substitution (Change of Variables)

Suppose you want to compute the integral $A = \int \frac{1}{\sqrt{4+9x^2}} dx$ by making a change of variables (also called a substitution). Actually, Maple can integrate this without changing variables, but it's easier to understand a technique when it is demonstrated on an easy example.

To do this by hand, you first need to identify a relation between the old variable x and a new variable. In this case, take

$$3x = 2 \tan \theta$$

where the new variable is θ . Compute the differential and solve for dx . You now have

$$3 dx = 2 \sec^2 \theta d\theta$$

$$dx = \frac{2}{3} \sec^2 \theta d\theta$$

Next, substitute x and dx into the integral to get

$$A = \int \frac{2 \sec^2 \theta}{3\sqrt{4 + 4 \tan^2 \theta}} d\theta = \frac{1}{3} \int \sec \theta d\theta = \frac{1}{3} \ln |\sec \theta + \tan \theta| + C$$

Finally, substitute back to get

$$A = \frac{1}{3} \ln \left| \frac{\sqrt{4 + 9x^2}}{2} + \frac{3x}{2} \right| + C$$

To do this computation using Maple, first enter the integral and the equation relating the old and new variables.

```
> A:=Int(1/sqrt(4+9*x^2),x);
```

$$A := \int \frac{1}{\sqrt{4 + 9x^2}} dx$$

```
> eq:=3*x=2*tan(theta);
```

$$eq := 3x = 2 \tan(\theta)$$

(Check you typed them correctly.) Then use the `changevar` command in the `student` package, which has three arguments: the first argument is the relation between the old and new variables; the second argument is the integral; and the third argument is the name of the new variable. So load the `student` package, execute the change of variables, and evaluate the integral (using `value`).

```
> with(student):
```

```
> changevar(eq,A,theta); Atemp:=value(%);
```

$$\int \frac{1}{3} \frac{2 + 2 \tan(\theta)^2}{\sqrt{4 + 4 \tan(\theta)^2}} d\theta$$

$$Atemp := \frac{1}{3} \operatorname{arcsinh}(\tan(\theta))$$

Finally, to substitute back, solve for the new variable and use the `eval` command.

```
> Avalue:=eval(Atemp, theta=arctan(3*x/2));
```

$$Avalue := \frac{1}{3} \operatorname{arcsinh}\left(\frac{3x}{2}\right)$$

This is the answer up to an additive constant. Maple's answer looks different from the one we obtained by hand, but the `convert` command shows that they are equivalent.

```
> convert(Avalue,ln);
```

$$\frac{1}{3} \ln\left(\frac{3x}{2} + \frac{\sqrt{4 + 9x^2}}{2}\right)$$

(Maple uses the formula $\operatorname{arcsinh}(w) = \ln(\sqrt{w^2 + 1} + w)$ for the conversion.) However, as with any indefinite integral, check the answer by differentiating the results. Thus,

```
> diff(Avalue,x);
```

$$\frac{1}{\sqrt{4+9x^2}}$$

gives back the original integrand in A .

If, on applying the `changevar` command, the integral becomes more complicated, go back and try a different substitution or a different integration trick.

You may ask why the `changevar` command is used when the commands

```
> A:=Int(1/sqrt(4+9*x^2),x); value(%);
```

$$A := \int \frac{1}{\sqrt{4+9x^2}} dx$$

$$\frac{1}{3} \operatorname{arcsinh}\left(\frac{3x}{2}\right)$$

will give the same result. The answer is that there are integrals that Maple cannot compute directly. Then an intelligent human-computer interaction may produce a result that Maple could not get on its own. For examples of such integrals, see the exercises.

6.4 Integration by Parts

Suppose you want to compute the integral $\int x \sin x \, dx$ by the method of integration by parts. The first thing to do is to identify u and dv . In this case take

$$u = x \quad \text{and} \quad dv = \sin x \, dx$$

Then compute du and v :

$$du = dx \quad \text{and} \quad v = -\cos x$$

The integration by parts formula $\int u \, dv = uv - \int v \, du$ gives

$$\int x \sin x \, dx = -x \cos x + \int \cos x \, dx = -x \cos x + \sin x + C$$

To do this computation using Maple, first enter the integral.

```
> A:=Int(x*sin(x),x);
```

$$A := \int x \sin(x) \, dx$$

(Check you typed it correctly.) Then use the `intparts` command in the `student` package, which has two arguments: the first argument is the integral and the second argument is the part of the integrand that will be taken as u . So, load the `student` package, execute the integration by parts, and evaluate the integral (using `value`).


```
> with(student):
> intparts(A,x); Aparts:=value(%);
```

$$-x \cos(x) - \int -\cos(x) dx$$

$$Aparts := -x \cos(x) + \sin(x)$$

Again, we could have combined these commands as

```
> value(intparts(Int(x*sin(x),x),x));
-x cos(x) + sin(x)
```

but it is better to work the problem in steps in order to verify that the integral is entered correctly.

If the integral becomes more complicated on applying the `intparts` command, go back and try a different u or a different integration technique. Alternatively, it may be necessary to integrate by parts more than once. Here is an example:

```
> B:=Int(x^2*sin(x),x);
```

$$B := \int x^2 \sin(x) dx$$

```
> B1:=intparts(B,x^2);
```

$$B1 := -x^2 \cos(x) - \int -2x \cos(x) dx$$

```
> intparts(B1,x); value(%);
```

$$-x^2 \cos(x) + 2x \sin(x) + \int -2 \sin(x) dx$$

$$-x^2 \cos(x) + 2x \sin(x) + 2 \cos(x)$$

6.5 Integration using Partial Fractions

Maple computes partial fractions by using the `convert` command with the `parfrac` option. This requires three arguments: a rational function, the keyword `parfrac` and the independent variable.

For example, to evaluate $\int \frac{x^2 - 3x + 1}{x^3 + x^2 - 2x} dx$, define the integrand

```
> f:=(x^2-3*x+1)/(x^3+x^2-2*x);
```

$$f := \frac{x^2 - 3x + 1}{x^3 + x^2 - 2x}$$

Find its partial fraction expansion

```
> fpar:=convert(f,parfrac,x);
```

$$fpar := \frac{11}{6(x+2)} - \frac{1}{3(x-1)} - \frac{1}{2x}$$

Construct the integral and find its value

```
> Int(fpar,x); value(%);
```

$$\int \frac{11}{6(x+2)} - \frac{1}{3(x-1)} - \frac{1}{2x} dx$$

$$\frac{11}{6} \ln(x+2) - \frac{1}{3} \ln(x-1) - \frac{1}{2} \ln(x)$$

Note that Maple omits the *absolute values* in its answer. It simply gives an antiderivative!

Of course, Maple could compute this directly

```
> Int(f,x); value(%);
```

$$\int \frac{x^2 - 3x + 1}{x^3 + x^2 - 2x} dx$$

$$\frac{11}{6} \ln(x+2) - \frac{1}{3} \ln(x-1) - \frac{1}{2} \ln(x)$$

but it may be useful to be able to check intermediate results in your hand computations using partial fractions.

6.6 Approximate Integration

When you can't compute an integral exactly, you may need to approximate it numerically using a left Riemann sum, a right Riemann sum, a midpoint Riemann sum, the trapezoid rule or Simpson's rule. At the beginning of this chapter, we discussed Maple's commands for left and right Riemann sums. In this section, we also discuss Maple's commands for midpoint Riemann sums, the trapezoid rule and Simpson's rule. We start by loading the `student` package

```
> with(student):
```

which contains the Maple commands `leftsum`, `rightsum`, `middlesum`, `trapezoid` and `simpson`. Let's use these commands to approximate

$$\int_0^1 \sin(x^2 + 1) dx$$

First define the function

```
> f:=x->sin(x^2+1);
```

$$f := x \rightarrow \sin(x^2 + 1)$$

The left, right and midpoint Riemann sums, trapezoid rule and Simpson's rule with 10 intervals are computed as follows:

```
> leftsum(f(x),x=0..Pi,10); l:=evalf(%);
```

$$\frac{1}{10} \pi \left(\sum_{i=0}^9 \sin\left(\frac{i^2 \pi^2}{100} + 1\right) \right)$$

$$l := 1.173839061$$

```
> rightsum(f(x),x=0..Pi,10); r:=evalf(%);
```

```

1/10 * pi * (sum_{i=1}^{10} sin(i^2 * pi^2 / 100 + 1))
r := 0.5978131996
> middlesum(f(x), x=0..Pi, 10); m:=evalf(%);
1/10 * pi * (sum_{i=0}^9 sin((i + 1/2)^2 * pi^2 / 100 + 1))
m := 0.8977656549
> trapezoid(f(x), x=0..Pi, 10); t:=evalf(%);
pi/30 * (sin(1) + sin(pi^2 + 1) + 4 * (sum_{i=1}^5 sin((2i-1)^2 * pi^2 / 100 + 1)) + 2 * (sum_{i=1}^4 sin(i^2 * pi^2 / 25 + 1)))
t := 0.8858261305
> simpson(f(x), x=0..Pi, 10); s:=evalf(%);
1/30 * pi * (sin(1) + sin(pi^2 + 1) + 4 * (sum_{i=1}^5 sin((2i-1)^2 * pi^2 / 100 + 1)) + 2 * (sum_{i=1}^4 sin(i^2 * pi^2 / 25 + 1)))
s := 0.9058553279

```

The last three are fairly close to Maple's result

```
> Int(f(x), x=0..Pi); i:=evalf(%);
```

$$\int_0^\pi \sin(x^2 + 1) dx$$

```
i := 0.8934801800
```

The error using Simpson's rule is

```
> E[S] := s - i;
```

```
E_S := 0.0123751479
```

Is this within the bound determined by the Simpson's rule error formula? Recall this formula says

$$|E_S| \leq \frac{K(b-a)^5}{180n^4} \quad \text{where } K \geq |f^{(4)}(x)| \quad \text{for all } a \leq x \leq b$$

Here:

```
> n:=10; (a,b):=(0,Pi);
```

```
n := 10
```


```
a, b := 0, pi
```

To find K , we compute $f^{(4)}(x)$ (You wouldn't want to do this by hand!) and plot its absolute value:

```
> d4f := (D@@4)(f);
```

```
d4f := x -> 16 sin(x^2 + 1) x^4 - 48 cos(x^2 + 1) x^2 - 12 sin(x^2 + 1)
```

```
> plot(abs(d4f(x)),x=0..Pi);
```



The maximum occurs at $x = \pi$. So we define K as

```
> K:=evalf(abs(d4f(Pi)));
      K := 1474.771668
```

So the error from Simpson's rule is bounded by

```
> error_bound:=evalf(K*(b-a)^5/(180*n^4));
      error_bound := 0.2507273118
```

Yes, the error we found $E_S = .0123751482$ is less than this error bound.

OK, now let's use the error bound for Simpson's rule to determine a value of n for which Simpson's rule approximates the integral within 10^{-5} . We unassign n and solve the inequality $E_S < 10^{-5}$.

```
> n:='n':
> eq:=evalf(K*(b-a)^5/(180*n^4)=10^(-5));
      eq :=  $\frac{2507.273118}{n^4} = 0.00001000000000$ 
> fsolve(eq,n=0..infinity);
      125.8346980
```

Thus n must be 126 or larger. Consequently, if we apply Simpson's rule with $n = 126$ then the result should have an error less than 10^{-5} .

```
> s:=evalf(simpson(f(x),x=0..Pi,126));
      s := 0.8934803287
> E[S]:=s-i;
      E_S := 0.1487 10-6
```

It does!

Remark: Maple uses a much more sophisticated method for numerically approximating integrals but the midpoint rule or Simpson's rule is usually sufficient.

6.7 Summary

- Load the student package and use `leftbox` and `rightbox` to help visualize rectangles in Riemann sums. Here are the steps:

```
with(student); ..... Load the student package.
f:=x-> expr; ..... Define the function.
leftbox(f(x), x=a..b,n); . Plot the left Riemann sum.
rightbox(f(x), x=a..b,n); Plot the right Riemann sum.
```

- Use Maple to set up a Riemann sum over the interval from a to b using n rectangles. Compute the area as the limit of the Riemann sums. Here are the steps:

```
f:=x-> expr; ..... Define the function.
Dx:=(b-a)/n; ..... Compute the rectangle width,
                    with a specific number n.
Sum(f(a+i*Dx)*Dx, i=1..n);
  area:=value(%); ..... Compute the Riemann sum.
Limit(area,n=infinity);
  value(%); ..... Compute the limit, which is the area.
```

- Set up Riemann sums with `leftsum` and `rightsum` which are in the `student` package.

```
leftsum(f(x), x=a..b,n); . Set up a left Riemann sum.
rightsum(f(x), x=a..b,n); Set up a right Riemann sum.
```

- Compute indefinite and definite integrals using `Int` and `value`

```
Int(f(x),x); value(%); ..... Compute an indefinite integral.
Int(f(x),x=a..b); value(%); Compute a definite integral.
```

- Recall that *Maple does not add a constant of integration.*

- Maple will only integrate an expression. If f is a Maple function, it must be evaluated to $f(x)$ before integrating.

- The Maple commands `changevar`, `intparts` and `convert/parfrac` can be used to perform integration by substitution, integration by parts and integration using partial fractions. Note `changevar` and `intparts` are in the `student` package. These commands can be used to help Maple calculate complicated integrals in closed form.

- Approximate definite integrals using `leftsum`, `rightsum`, `middlesum`, `trapezoid` and `simpson` which are in the `student` package. Perform the sums using `value(%)`.

```
leftsum(f(x), x=a..b,n); .. Set up a left Riemann sum.
rightsum(f(x), x=a..b,n); . Set up a right Riemann sum.
middlesum(f(x), x=a..b,n); Set up a midpoint Riemann sum.
trapezoid(f(x), x=a..b,n); Set up a trapezoid rule sum.
simpson(f(x), x=a..b,n); .. Set up a Simpson's rule sum.
```

6.8 Exercises

1. Load the `student` package via `with(student);`
 - (a) Use the `leftbox` command with 9 rectangles to display the left Riemann sum approximation to $\int_0^2 e^{x^2/2} dx$. Does this set of rectangles overestimate or underestimate the integral?
 - (b) Set up and evaluate the left Riemann sums to approximate the integral with $n = 9, 27, 81$ and 243 rectangles using the `leftsum` command.
 - (c) Compute the integral using Maple's `Int` and `evalf(%)` commands as done in the text.
 - (d) Compare the values of the Riemann sums in (b) to the value of the integral in (c). (Subtract the exact value from the estimates.) As the number of rectangles triples, what happens to the difference between the value of the Riemann sum and the value of the integral?
2. Repeat Exercise 1 for right Riemann sums, using the `rightbox` and `rightsum` commands.
3. Repeat Exercise 1 for midpoint Riemann sums, using the `middlebox` and `middlesum` commands.
4. Repeat Exercises 1, 2 and 3 with the function $f(x) = \frac{1-x^4}{x^2+1}$ using the interval $[0, 1]$.
5. Use Maple to evaluate the following integrals. Check (b) and (c) by differentiating.

(a) $\int_1^3 x^4 \sqrt{x^2+1} dx$

(b) $\int \sin^2(x) dx$

(c) $\int x \cos^4(x) dx$

6. There are two ways to evaluate the integral $\int \sec^4(x) \tan(x) dx$ by u -substitution. The first is to let $u = \sec(x)$ and the second is to let $u = \tan(x)$. Based on Maple's answer to this integral, (using `Int` and `value`) which substitution is it using? For each substitution, use `changevar`, and `value` and use `eval` to substitute back. These two methods of substitution lead to apparently different answers. Are they, in fact, the same? If they are different, then how do you explain the fact that they are answers to the same problem? (Subtract the answers and `simplify`.)

7. Express each of the following limits as an integral and then use Maple to evaluate. (The first two integrals are provided for you.)

$$(a) \lim_{n \rightarrow \infty} \sum_{i=1}^n \frac{1}{n} \sqrt{4 + \frac{i^2}{n^2}} = \int_0^1 \sqrt{4 + x^2} dx$$

$$(b) \lim_{n \rightarrow \infty} \sum_{i=1}^n \frac{3}{n} \sqrt{9 + \frac{9i^2}{n^2}} = \int_0^3 \sqrt{9 + x^2} dx$$

$$(c) \lim_{n \rightarrow \infty} \sum_{i=1}^n \frac{\pi}{n} \cos^4 \left(\frac{i\pi}{n} \right)$$

$$(d) \lim_{n \rightarrow \infty} \sum_{i=1}^n \frac{3}{n} \left(\left(2 + \frac{3i}{n} \right)^4 + 6 \left(2 + \frac{3i}{n} \right) \right)$$

8. This problem is a continuation of Exercise 9 in Chapter 5. In that exercise, the northern boundary of a plot of land is described by the following table (each unit represents 100 feet).

x -value	0.0	1.0	2.0	3.0	4.0
y -value	3.2	1.5	2.6	3.2	2.7

where the x -axis represents the (straight) southern boundary with the origin located at its western corner. (The eastern and western boundaries are also straight.) Use these data to compute an approximation to the area of this plot of land using trapezoids.

9. Now do Exercise 9 from Chapter 5 or load the answer from a file if you have already done that exercise. Use the quadratic and cubic functions described therein to evaluate the area of this plot of land.
10. *Area of Texas.* As in Exercise 2 in Chapter 3, the northern and southern boundaries of the state of Texas are given by the following data.
`north:=[[0,0],[3,0],[3,4.5],[6,4.5],[6,2.2],[7,2.1],[8,1.8],[9,1.9],[10,1.8],[11,1.7],[11,-2.2]];`
`south:=[[0,0],[1,-1.1],[2,-2.5],[3,-2.9],[4,-2.3],[5,-2.8],[6,-4.4],[7,-5.8],[8,-6.1],[9,-3.3],[10,-2.8],[11,-2.2]];`
 (Enter these in Maple.) Here, the origin is the westernmost corner of Texas (near El Paso) and the x -axis is the extension of the east-west border between New Mexico and Texas. Each unit represents approximately 69 miles. Use these data to approximate the area of Texas by computing a left Riemann sum formed from rectangles whose widths are 1 unit and whose heights are determined by the second coordinates of the data.
 HINT: To refer to the entries in a list, use square brackets []. For example, `south[3]` refers to the point `[2, -2.5]` and `south[3][2]` refers to the second entry of this point, `-2.5`. So to (symbolically) sum up all second entries of the points on this list, enter `Sum(south[i][2], i=1..11)`; Then, to actually compute the sum, enter `value(%)`;

11. *Area of Texas, Two.* Improve your estimate from Exercise 10 by using trapezoids instead of rectangles.
12. *Area of Texas, Three.* Improve your estimate from Exercise 10 by using the quadratic and cubic curves found in Exercise 10 of Chapter 5.
13. Evaluate the integral $\int \frac{x^5 \arctan(x^2)}{(1+x^4)^3} dx$. First try the `value` command by itself. The answer is not useful because it is given in terms of complex numbers. Then use the `changevar` command with $u = \arctan(x^2)$, followed by `value`, `eval`, and `simplify`. Be sure to check your answer using `diff`, and `simplify`.
14. Evaluate the integral $\int x^2 \ln(x + \sqrt{1+x^2}) dx$. First try the `value` command by itself. Maple gives up. Then use the `intparts` and `value` commands with $u = x^2$. Can Maple do it now? Finally, use the `intparts` and `value` commands with $u = \ln(x + \sqrt{1+x^2})$. Can Maple do it now? Be sure to check the answer.
15. Find the partial fraction expansion for $f = \frac{x^6 + 3x^2 - 1}{(x+2)^2(x^2+4)^2}$ and use it to evaluate $\int \frac{x^6 + 3x^2 - 1}{(x+2)^2(x^2+4)^2} dx$.
16. Compute $\int \frac{x^3}{x^8 + 1} dx$
17. Compute $\int \frac{\sin(\sqrt{x})}{2\sqrt{x}} dx$
18. Compute $\int \cos^2(x) \sin^4(x) dx$
19. Compute $\int_{-1}^3 \frac{x^5 + x}{\sqrt{1+x^2}} dx$
20. Compute $\int_{\pi/4}^{3\pi/4} \sqrt{\csc^2(x) - 1} dx$
21. Repeat Exercise 1 b,c,d using the trapezoid rule instead of left Riemann sums for the integral $\int_0^2 e^{x^2/2} dx$.
Explain carefully why we cannot repeat the exercise using Simpson's rule.
22. Use the error estimate for Simpson's rule to determine a value for n so that if Simpson's rule is used to approximate the integral $\int_0^2 e^{x^2/2} dx$, then the value is accurate to 15 decimal places. Calculate the Simpson's rule approximation for this n and check its accuracy.